計算機支援数学

１月２１日，２８日，２月４日分プリント
（担当：佐伯　修）

以下，*Mathematica* を使って平面図形，特に曲線を描く方法について簡単に解説する．詳細については，このプリントの最後にある [4] や，*Mathematica* のヘルプを参照して欲しい．[5] や [1, 第 I 部] にもかなりの事柄が解説されているので，参考になるであろう．ちなみに，１月２１日，２８日の講義で，平面曲線の微分幾何学の基礎的事項について解説するが，それらについては [3] や，[1, 第 I 部] の第１章を参照していただきたい．なお，閉曲線の頂点の個数に関する四頂点定理については，梅原先生による解説 [2] が参考になる．

なお，最終的には，§7 のレポート問題ができれば良いので，*Mathematica* に精通している学生は，直接レポート問題にチャレンジしても構わなし，精通していない学生も，以下をすべて読む必要はない．

1　関数のグラフ

まず，関数のグラフを描くことから始めよう．`Plot[f[x],{x,a,b}]` と入力すると，$y = f(x)$ のグラフが $a \leq x \leq b$ の範囲で描かれる．

```
In[1]:= f[x_] := x^2
In[2]:= Plot[f[x], {x, -1, 1}]
```



なお，`In[1]` では，`f[x]` を定義しているが，`f[x]` ではなく，`f[x_]` となっていることに注意して欲しい．関数をいちいち定義せず，直接入力することももちろん可能である．つまり，

```
In[3]:= Plot[x^2, {x, -1, 1}]
```

としても同じ結果が出力される．

しかし，いずれにしても上のようにすると，縦と横の比が通常と異なってしまう.（横幅を縦幅で割った比 AspectRatio は，デフォルト（初期値）では，黄金比の逆数 1/GoldenRatio となっている．ちなみに

```
In[4]:= N[GoldenRatio]
Out[4]= 1.61803
```

とすれば，黄金比の値を返してくる.）縦横比を１：１にしたいときは

```
In[5]:= Plot[f[x], {x, -1, 1}, AspectRatio -> Automatic]
```

とすれば良い．すると下図のように出力される．



なお，その他いろいろなオプションが用意されている．たとえば，

```
In[6]:= Plot[f[x], {x, -1, 1}, AspectRatio -> Automatic,
  PlotStyle -> {Thickness[0.01], Dashing[{1/50, 1/50}], Hue[1]}]
```

とすると，下図のように，線を太くしたり，点線にしたり,（プリントでは印刷の都合上，カラーにはなっていないけれども）線を赤くしたりできる.（数字を変えて，各自実験してみて欲しい.）



その他のオプションについては，たとえば [1, p. 192] などを参照していただきたい．

2 パラメータ表示された平面曲線

次に，パラメータ表示の与えられた平面曲線の描き方を紹介しよう．

```
In[7]:= ParametricPlot[{t^3, t^2}, {t, -10, 10},
AspectRatio -> Automatic]
```



これは，$\gamma(t) = (t^3, t^2)$ $(-10 \leq t \leq 10)$ なるパラメータ表示の与えられた平面曲線である．一つの座標平面上に複数個の曲線を描くには，

```
In[8]:= ParametricPlot[{{t, t^2}, {t, t^3}}, {t, -1, 1}]
```



のように，複数個の曲線をリストとして書くと良い．ただし，これではパラメータの範囲が共通となるので，異なる範囲の場合は，

```
In[9]:= Show[ParametricPlot[{t^2, t^3}, {t, -1, 1}],
  ParametricPlot[{t, t}, {t, -2, 2}],
  AspectRatio -> Automatic]
```



のように Show というコマンドを用いると良い．関数のグラフを描く場合も同様に，

```
In[10]:= Plot[{x^2, x^3}, {x, -1, 1}]
```

などとすれば良い.

練習問題 1．次の関数のグラフをそれぞれ描け．(x の範囲は各自で適当に選んで良い.) また，それらを一つの座標平面にまとめて描け.

(1) $f_1(x) = \cos 2\pi x$　(2) $f_2(x) = e^{x/3}$　(3) $f_3(x) = \log(x+1)$

答の例：



練習問題２．次の平面曲線をそれぞれ描け．また，それらを一つの平面上にまとめて描け．

(1) $\gamma_1(t) = \left(\frac{1}{3}\cos t, \frac{1}{2}\sin t\right) \qquad (0 \leq t \leq 2\pi)$

(2) $\gamma_2(t) = (\cos t/(1+\sin^2 t), \cos t \sin t/(1+\sin^2 t)) \qquad (0 \leq t \leq 2\pi)$

(3) $\gamma_3(t) = (\log t \cos t, \log t \sin t) \qquad (\pi \leq t \leq 9\pi)$

答：



3　接線と法線

次に曲線の接線を描くプログラムの例を紹介しよう．

```
In[11]:= tangentline[g_][a_, b_, c_, t1_, t2_] :=
  Module[{t, dg},
    dg = g';
    Show[ParametricPlot[Evaluate[g[t]], {t, a, b}],
      Graphics[Line[{g[c] + t1 dg[c], g[c] + t2 dg[c]}]],
      AspectRatio -> Automatic]
    ]
```

これは，パラメータ表示が $g(t)$ で与えられている平面曲線（ということは，$g(t)$ は２つの成分からなるリストとして与えられている，ということである）を，$a \leq t \leq b$ の範囲で描き，その $t = c$ における接線

$$t \mapsto g(c) + tg'(c)$$

を $t_1 \leq t \leq t_2$ の範囲で，同じ座標平面上に描くプログラムである．これを実際に使うと，たとえば以下のようになる．

```
In[12]:= g1[t_]:={t,t^2}
In[13]:= tangentline[g1][-2, 2, 1, -2, 1]
```



少し説明を加えておこう．Module は，変数をそこの中だけで局所的に使う場合に便利である．ここでは t，dg を使うので，最初にそれを宣言している．次の dg = g' で，g の各成分関数を微分したものを dg としている．dg もリストとなる．ParametricPlot の中に現れる Evaluate は，無くても良いが，これを入れておくことにより，計算効率が上がることが良くあるし，場合によってはこれを入れないとプログラムがうまく動かないことが良くある.「おまじない」のようなものと思って良い．Line はそこに与えられた 2 点を結ぶ線分を出力する．なお，その前にある Graphics は，グラフィックスオブジェクトを作る命令であり，これも「おまじない」のようなものである.

曲線の法線を描くためには，接線を（左に）90° 回転する作業が必要になる．これは，接ベクトルのそれぞれの成分を用いて直接的に計算させても良いが，回転行列を用いる方法の方が汎用性があるので，その方法を紹介する.

```
In[14]:= r[t_] := {{Cos[t], -Sin[t]}, {Sin[t], Cos[t]}}
In[15]:= r[0]
Out[15]= {{1, 0}, {0, 1}}
In[16]:= r[Pi/2].{1, 1}
Out[16]= {-1, 1}
```

行列は，横ベクトルを並べることによって表すことに注意して欲しい．In[14] で回転角 t の回転行列を定義している．回転角 0 の行列はもちろん単位行列である（In[15]，Out[15]）．ベクトル $(1,1)$ を左に $\pi/2$ 回転すると，$(-1,1)$ となる（In[16]，Out[16]）．なお，行列の積は In[16] のように，真中にピリオドを打って表すが，二つともベクトルのときは内積を表すので注意して欲しい．

では，曲線の法線を描くプログラムを作ってみよう．

```
In[17]:= normalline[g_][a_, b_, c_, t1_, t2_] :=
  Module[{t, dg, r, n},
    dg = g';
    r := {{0, -1}, {1, 0}};
    n := r. g'[c];
    Show[ParametricPlot[Evaluate[g[t]], {t, a, b}],
      Graphics[Line[{g[c] + t1 n, g[c] + t2 n}]],
      AspectRatio -> Automatic]
    ]
In[18]:= g1[t_] := {t, t^2}
In[19]:= normalline[g1][-2, 2, 1, -1, 2]
```



詳細については解説しないので，各自で考えて欲しい．

練習問題３．（１）曲線上の与えられた点における接線と法線を同時に描くプログラムを作れ．

（２）接線を，与えられた角度だけ回転してできる直線を描くプログラムを作れ．

次に，曲線の接線族を描くプログラムを紹介しよう．

```
In[20]:= tangentfield[f_][a_, b_, d_, i_, j_] :=
  Module[{fa, fb, u, v, vectors, t},
    fa[u_] := f[u] + i*f'[u];
    fb[v_] := f[v] + j*f'[v];
    vectors = Table[{fa[t], fb[t]}, {t, a, b, d}];
    Show[Graphics[Map[Line, vectors],
        AspectRatio -> Automatic, Axes -> False],
      ParametricPlot[Evaluate[f[t]], {t, a, b},
        AspectRatio -> Automatic]
      ]
    ]
```

これは与えられた平面曲線 $f(t)$ $(a \leq t \leq b)$ を描き，t の値を a から d ごとに増やしていったときの点 $f(a+nd)$ $(n=0,1,\ldots,[(b-a)/d])$ における接線

$$f(a+nd)+tf'(a+nd)$$

を，$i \leq t \leq j$ の範囲で次から次に描くプログラムである（$[(b-a)/d]$ はガウス記号を表す）．なお，`Axes -> False` というオプションは，座標軸を描かないために用いている．たとえば，

```
In[21]:= tangentfield[g1][-2, 2, 0.5, -1, 1]
```

とすると，下図のように出力される．



少しプログラムの説明をしておこう．Table は，繰り返しの作業を行うときに使う．上では，接線の端点の座標を組にしたリストを，次から次に生成させるために使っている．すべての接線の端点をリストにしたものが vectors として作られる．Map は，リストを一度に Line の中に入れるために使われている．

4　曲率

次に平面曲線の曲率を計算してみよう．講義で紹介したように，パラメータ表示の与えられた平面曲線

$$\gamma(t) = (x(t), y(t))$$

の曲率関数 $\kappa(t)$ は，

$$\begin{aligned}\kappa(t) &= \frac{\dot{x}(t)\ddot{y}(t) - \dot{y}(t)\ddot{x}(t)}{(\dot{x}(t)^2 + \dot{y}(t)^2)^{3/2}} \\ &= \frac{\det\begin{pmatrix}\dot{\gamma}(t) \\ \ddot{\gamma}(t)\end{pmatrix}}{\left(\sqrt{\dot{\gamma}(t)\cdot\dot{\gamma}(t)}\right)^3}\end{aligned}$$

で与えられるのであった．たとえば，In[18] で定義した曲線 $g_1(t)$ の曲率を計算すると，

```
In[22]:= Det[{g1'[t], g1''[t]}]/Sqrt[g1'[t].g1'[t]]^3 // Simplify
```

に対して

$$\frac{2}{(1+4t^2)^{3/2}}$$

という答が返ってくる．ここで，ベクトルの内積は，

```
g1'[t].g1'[t]
```

のように，二つのベクトルを表すリストの間にピリオドを打って表すことに注意して欲しい．

練習問題４．（１）パラメータ表示の与えられた平面曲線の曲率関数のグラフを描くプログラムを作れ．

（２）曲率関数の導関数のグラフを描くプログラムを作れ．

（３）いくつかの曲線について，上の（１）、（２）のプログラムを実行し，それらの曲線の頂点（曲率関数の極値を与える曲線上の点）の個数を求めよ．

5　弧長

曲線の弧長を求めるプログラムを紹介しよう．弧長は，曲線の速度ベクトルの長さを積分して得られることを思い出そう．

```
In[23]:= arclength[f_][a_, b_] :=
  Module[{v, t},
    v[t] := Sqrt[f'[t].f'[t]];
    NIntegrate[Evaluate[v[t]], {t, a, b}]
    ]
In[24]:= g2[t_] := {Cos[t], Sin[t]}
In[25]:= arclength[g2][0, Pi]
Out[25]= 3.14159
```

NIntegrate は定積分の値を求めるコマンドである．不定積分の形で式として求めたいときには Integrate を用いる．

```
In[26]:= arclength1[f_] :=
  Module[{v},
    v[t] := Sqrt[f'[t].f'[t]];
    Integrate[Evaluate[v[t]], t]
    ]
In[27]:= g3[t_] := {t, t}
In[28]:= arclength1[g3]
```

こうすると，答として $\sqrt{2}t$ が返ってくる．なお，In[26] では，Module の中の最初に，局所変数として t があえて書かれていないことに注意して欲しい．書いてもプログラムとしては動くが，返されてくる関数の変数が変な形になってしまう．

練習問題 5．いろいろな平面曲線の弧長を，上のプログラムを使って計算せよ．たとえば，楕円の弧長を計算させると，どんな式が出てくるか？

6　平面曲線の陰関数表示

平面曲線が，ある 2 変数関数 $F(x, y)$ を用いて

$$\{(x, y) \in \mathbf{R}^2 \mid F(x, y) = 0\}$$

の形に書けるとき，これを曲線の**陰関数表示**という．

陰関数表示された曲線も *Mathematica* を用いて描くことができる．たとえば，

```
In[29]:= Needs["Graphics`ImplicitPlot`"]
In[30]:= ImplicitPlot[x^2 + y^2 - 1 == 0, {x, -1, 1}]
```

とすると，下図のように出力される．



ImplicitPlot という命令を使うには，Needs という命令を用いて，パッケージより読み込まなくてはならない．ImplicitPlot の場合は，Graphics というカテゴリに属するので，Graphics`ImplicitPlot` を入力する．ここで，クォーテーションマークは，普通の ' ではなくて，逆向きの ` であるので，注意して欲しい．また，Needs の中身にはダブルクォーテーションマークが必要であることにも注意して欲しい．

次の例のように，特異点を持った曲線も表せる．

```
In[31]:= ImplicitPlot[x^2 - y^3 == 0, {x, -1, 1}]
```



練習問題 6．楕円のパラメータ表示と陰関数表示を求め，それらを用いて楕円を描いてみよ．まったく同じ絵となるか？

7 レポート問題

以下の中から**一題以上**を選んで解答し，それを佐伯まで**電子メイルで送る**こと．形式は，メイル本文に *Mathematica* のプログラムを貼り付ける形でも良いし，プ

ログラムを添付ファイルとして送ってくれても良い.（なお，プログラムの説明をつけることが望ましい.）佐伯のアドレスは

saeki@math.sci.hiroshima-u.ac.jp

である．送ってもらったものを私のパソコンで走らせて，うまくゆくかどうか確認した上で成績を評価する．提出期限は２月８日（金）午後５時とする．

なお提出の際には，計算機支援数学の講義全体に関する感想・意見・質問・苦情などを忘れずに添付すること（特に佐伯担当分に関するものでも良いし，そうでなくても良い）．その内容は成績評価の対象とはしないので，今後の参考のために，学生諸君の率直な声を聞かせて欲しい.

なお，レポート問題のうち，解答自体がプログラムでないものについては，レポート用紙などに書いて提出しても構わないし，TEX などで書いて電子メイルで送ってもらっても構わない．紙に書いたものの提出先は数学事務室で，提出期限は上記と同じとする.

なお，どうしても電子メイルでのレポート提出が困難である学生は，早めに佐伯まで申し出ること.

1．（１）（特異点を持たない）平面曲線 $\gamma(t)$ が与えられたときに，その単位速度ベクトル $\mathbf{t}(t)$，法線ベクトル $\mathbf{n}(t)$ を（式として）求めるプログラムを作れ.

（２）$\mathbf{n}(t)$ を新しい平面曲線と考えたとき，その弧長 K を数字として求めるプログラムを作れ（この K のことを曲線の**全曲率**という）.

（３）いくつかの閉曲線や卵形線について（２）のプログラムを使って K の値を求め，それらが常に 2π 以上であり，卵形線についてはちょうど 2π となることを確かめよ.（実は，閉曲線については常に $K \geq 2\pi$ であって，等号成立は卵形線のとき，かつそのときに限ることが知られている．詳細はたとえば [3, 第 1 章，§3] を参照.）

2．（１）（特異点を持たない）平面曲線が与えられたとき，その曲率関数を求めるプログラムを作れ.

（２）曲率関数の積分値を求めるプログラムを作れ.

（３）いくつかの（特異点を持たない）閉曲線に対して（２）のプログラムを使って積分値を計算し，それらがすべて 2π の整数倍となっていることを確かめよ．この整数のことを，閉曲線の**回転数**と言う.

（４）回転数を求めるプログラムを作れ.

3．（１）（特異点を持たない）平面曲線が与えられたとき，その曲線と，縮閉線と，法線族を同じ座標平面上に描くプログラムを作れ.（異なる線は，色や線の太さを異なるようにして，見やすくすることが望ましい.）

（２）上のプログラムを使って，縮閉線が法線族の包絡線であることが良くわかるような例を探せ.

（３）頂点がちょうど６つである単純閉曲線の例を探せ.

４．（１）与えられた（特異点を持たない）平面曲線 $\gamma(t)$ に対して，その曲線と，その平行曲線

$$P_\gamma(t) = \gamma(t) + a\mathbf{n}(t)$$

を描くプログラムを作れ.

（２）上の a をいろいろと変えて，平行曲線の族を同一座標平面上に自動的に描くプログラムを作れ.

（３）上のものに，さらに縮閉線を同時に描くプログラムを作れ.

（４）上のプログラムを使って，平行曲線の特異点が縮閉線の上にあることが良くわかる例を探せ.

５．（１）平面曲線の陰関数表示 $F(x,y)=0$ が与えられているとする．この曲線上の点 (x_0,y_0) における接線の方程式は,

$$\frac{\partial F}{\partial x}(x_0,y_0)(x-x_0) + \frac{\partial F}{\partial y}(x_0,y_0)(y-y_0) = 0$$

で与えられる．このことを示せ.

（２）点 (x_0,y_0) における曲線の法線ベクトルが,

$$\left(\frac{\partial F}{\partial x}(x_0,y_0), \frac{\partial F}{\partial y}(x_0,y_0)\right)$$

で与えられることを示せ.

（３）陰関数表示で与えられた曲線の，与えられた点における接線，法線を描くプログラムを作れ.

（ヒント：関数の偏微分は，*Mathematica* では次のように行う.

```
In[32]:= G[x_, y_] := x^2y^3 + 5x^3y^6
In[33]:= D[G[x, y], {x, 1}, {y, 2}]
Out[33]= 12 x y + 450 x^2 y^4
```

これは，$G(x,y)$ を x で 1 回，y で 2 回偏微分したものである.）

６．パラメータ表示された曲線 $\gamma(t) = (x(t), y(t))$ $(t \in I)$ を考える．今，$v \in S^1 = \{(x,y) \in \mathbf{R}^2 \mid x^2+y^2=1\}$ としたとき，$h: I \to \mathbf{R}$ を，内積を用いて,

$$h(t) = \gamma(t) \cdot v$$

で定義する．これを v 方向への**高さ関数**という.

（１）平面曲線 γ が弧長 s でパラメータづけられているとき，$s_0 \in I$ に対して次が成り立つことを示せ．

(a) $h'(s_0) = 0 \Longleftrightarrow v = \pm \mathbf{n}(s_0)$

(b) $h'(s_0) = h''(s_0) = 0 \Longleftrightarrow \kappa(s_0) = 0$ かつ $v = \pm \mathbf{n}(s_0)$

（２）上の (a), (b) を，曲線 γ と直線の接触の仕方の観点から幾何学的に解釈せよ．なお，上の (b) を満たす $s_0 \in I$（つまり，$\kappa(s_0) = 0$ となる s_0）に対して $\gamma(s_0)$ のことを，曲線の**変曲点**という．

（３）以下，曲線 γ は原点を通らないものとする．γ の各点における接線に対して，原点から引いた垂線の足の軌跡を，**垂足曲線**という．すると垂足曲線 $PE_\gamma(s)$ は，

$$PE_\gamma(s) = (\gamma(s) \cdot \mathbf{n}(s))\mathbf{n}(s)$$

で与えられる．これを示せ．

（４）垂足曲線の特異点は，もとの曲線 γ の変曲点に対応する．すなわち，

$$PE'_\gamma(s_0) = 0 \Longleftrightarrow \kappa(s_0) = 0$$

となる．これを示せ．

（５）与えられた曲線に対して，垂足曲線を描くプログラムを作れ．

（６）関数 $G : I \times (\mathbf{R}^2 - \{(0,0)\}) \to \mathbf{R}$ を

$$G(s, \mathbf{x}) = \gamma(s) \cdot \left(\frac{\mathbf{x}}{||\mathbf{x}||} \right) - ||\mathbf{x}||$$

で定義する．$s_0 \in I$ を固定したとき，方程式 $G(s_0, \mathbf{x}) = 0$ が定める曲線は，半径が $||\gamma(s_0)||/2$ で中心が $(1/2)\gamma(s_0)$ の円である．垂足曲線が，曲線族 $G(s, \mathbf{x}) = 0$ の包絡線であることを示せ．

（７）与えられた平面曲線，その垂足曲線，そして（６）の曲線族を同一平面上に描くプログラムを作り，垂足曲線が包絡線であることが良くわかる例を探せ．

７．幾何学に関するプログラムで，数学的に興味深いと思われるものを自由に作れ．それを使った具体例も添えること．

参考文献

[1] 泉屋周一，佐野貴志，佐伯修，佐久間一浩『幾何学と特異点』特異点の数理，第一巻，共立出版，２００１年．

[2] 梅原雅顕『四頂点定理について』数学 **50** (1998), 84–91.

[3] 小林昭七『曲線と曲面の微分幾何』（改訂版）裳華房，１９９５年．

[4] スティーブン・ウルフラム『MATHEMATICA ブック』（第４版）東京書籍，２０００年．

[5] Ｓ．ワゴン著，長岡亮介監訳『*Mathematica* で見える現代数学』ブレーン出版，１９９６年．